

```
1 ' -----
2 '       From the book Inside Active Directory, ISBN 0-201-61621-1
3 '       Copyright (C) 2002 by Addison-Wesley
4 '       Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim objDSE, strDefaultDN, strDN, objADObject, i
14 Dim objSecDesc, objDAcl, objACE, objExcel
15
16 Set objDSE = GetObject("LDAP://rootDSE")
17 strDefaultDN = "CN=Users," & objDSE.Get("defaultNamingContext")
18
19 strDN = InputBox("Enter the distinguished name of an object" & _
20 vbCrLf & "(e.g. " & strDefaultDN & ")", , strDefaultDN)
21
22 If strDN = "" Then WScript.Quit(0) 'user clicked Cancel
23
24 Set objADObject = GetObject("LDAP://" & strDN)
25
26 Set objSecDesc = objADObject.Get("ntSecurityDescriptor")
27 Set objDAcl = objSecDesc.DiscretionaryAcl
28
29 Set objExcel = WScript.CreateObject("Excel.Application")
30 objExcel.Visible = True
31 objExcel.Workbooks.Add
32 objExcel.ActiveSheet.Name = "ACEs"
33
34 objExcel.ActiveSheet.Range("A1").Activate
35 objExcel.ActiveCell.Value = "Trustee"
36 objExcel.ActiveCell.Offset(0,1).Value = "AccessMask"
37 objExcel.ActiveCell.Offset(0,2).Value = "AceFlags"
38 objExcel.ActiveCell.Offset(0,3).Value = "AceType"
39 objExcel.ActiveCell.Offset(0,4).Value = "Flags"
40 objExcel.ActiveCell.Offset(0,5).Value = "ObjectType"
41 objExcel.ActiveCell.Offset(0,6).Value = "InheritedObjectType"
42 objExcel.ActiveCell.Offset(1,0).Activate 'goto next line
43
44 i = 0
45 For Each objACE In objDAcl
46     objExcel.ActiveCell.Value = objACE.Trustee
47     objExcel.ActiveCell.Offset(0,1).Value = Hex(objACE.AccessMask)
48     objExcel.ActiveCell.Offset(0,2).Value = Hex(objACE.AceFlags)
49     objExcel.ActiveCell.Offset(0,3).Value = Hex(objACE.AceType)
50     objExcel.ActiveCell.Offset(0,4).Value = Hex(objACE.Flags)
51     objExcel.ActiveCell.Offset(0,5).Value = objACE.ObjectType
52     objExcel.ActiveCell.Offset(0,6).Value = objACE.InheritedObjectType
53     objExcel.ActiveCell.Offset(1,0).Activate 'goto next line
54 Next
55
```